CM0377 2000/2001 Lecture 11: Inference engines in Prolog

The backward chaining interpreter (Simpl.pl)

prove(Fact):-


if_then(Conditions,Fact),


infer_each(Conditions).

prove(Fact):-


ask(Fact).

infer_each([]).

infer_each([Fact|Rest]):-


prove(Fact),


infer_each(Rest).

ask(Fact):-


write('Is it true that '),


write(Fact), write('?'), nl,


read(Ans),


Ans=yes.

run:-


retractall(if_then(_,_)),


write('Select a knowledge base ...'),


read(KB),


load_files([KB],


   [compilation_mode(assert_all)]),


write('Type in a goal ...'),


read(Goal),


prove(Goal),


write('Proved!'), nl, !.

run:-


write('Not proved!'), nl.

Knowledge base (Simpl1.pl)

if_then(['you are fat','you do not eat'],'you are getting slimmer').

if_then(['you are fat','you eat wimpey carpet-burgers'],'you are in danger of going pop').

if_then(['you are fat','you are rich'],'you can become an american software big-shot').

Another knowledge base (Simpl2.pl)

if_then(['it has hair'],'it is mammal').

if_then(['it gives milk'],'it is mammal').

if_then(['it is mammal','it is tawny','it is carnivore','it has dark spots'],

   'it is a cheetah').

if_then(['it is mammal','it is tawny','it is carnivore','it has black stripes'],

   'it is a tiger').

Modified inference engine, with Certainty Factors (cf.pl)

prove(Fact, Cert):-


if_then(Conditions,Fact),


infer_each(Conditions, Cert).

prove(Fact, Cert):-


ask(Fact, Cert).

infer_each([], 1).

infer_each([Fact|Rest], Cert):-


prove(Fact, C1),


infer_each(Rest, C2),


min(C1, C2, Cert).

min(X, Y, X):- X < Y, !.

min(X, Y, Y).

ask(Fact, Cert):-


write('Please give certainty for '),


write(Fact), write('...'), nl,


read(Cert).

run:-


retractall(if_then(_,_)),


write('Select a knowledge base ...'),


read(KB),


load_files([KB],


   [compilation_mode(assert_all)]),


write('Type in a goal ...'),


read(Goal),


prove(Goal, Cert),


write('Certainty is '),


write(Cert), nl, !.

Forward chaining program (fwd.pl)

:-dynamic(fired/1).

:-dynamic(stopped/0).

:-dynamic(current_list/1).

go:-

  retractall(current_list(_)),

  retractall(stopped),

  assert(current_list([5,3,2,4,1])),

  current_list(X),

  write('Initial state:  '), write(X), nl,

  run_rules.

run_rules:-

  \+(stopped),!,

  conflict_set(CList),

  choose_rule(CList,Rule),

  fire_rule(Rule),

  run_rules.

run_rules.

conflict_set(CList):-

  findall(R, belongs_to_c_set(R), CList).

belongs_to_c_set(rule(N, Cond, Act)):-

  rule(N, Cond, Act),

  satisfy(Cond).

fire_rule(rule(N, Cond, Act)):-

  call(Act),!.

choose_rule([R|Ignore], R).

rule(1,[left_of(A,B),left_of(B,C),


bigger_than(A,C)], swap(A,C)).

rule(2,[left_of(A,B),bigger_than(A,B)],


swap(A,B)).

rule(3,[no_left_of_and_bigger],stop).

satisfy(Conditions):-

  satisfy_each(Conditions),!.

satisfy_each([]).

satisfy_each([H|T]):-

  call(H), satisfy_each(T).


left_of(A,B):-

  current_list(L),

  split(L,A,[B|_]).

bigger_than(A,B):-

  A>B.

no_left_of_and_bigger:-

  \+(some_left_of_and_bigger).

some_left_of_and_bigger:-

  left_of(A,B),

  bigger_than(A,B).

swap(A,B):-

  retract(current_list(L)),

  list_with_swap(L,A,B,New),

  assert(current_list(New)),

  write('Swap performed: '),

  write(New),nl.

stop:-

  assert(stopped).

list_with_swap([],_,_,[]).

list_with_swap([A|T],A,B,[B|T1]):-

  list_with_swap(T,A,B,T1).

list_with_swap([B|T],A,B,[A|T1]):-

  list_with_swap(T,A,B,T1).

list_with_swap([H|T],A,B,[H|T1]):-

  \+(H=A),\+(H=B),

  list_with_swap(T,A,B,T1).

split([H|T],H,T).

split([H|T],A,List):-

  split(T,A,List).

